
属性エクセル編集

■ やりたいこと

図上のブロックの属性情報をエクセルシートに書きだす。
書き出した内容をエクセルシート上で編集する。
編集内容を図に反映する。

設備NO ：ブロックの表す設備の識別番号
品名  ：ブロックの表す設備の一般的な名称
寸法  ：ブロックの表す設備の寸法を W1000xD500xH800 形式で記入
品番  ：ブロックの表す設備のメーカー品番

■ ブロック内属性定義の内容

■ エクセル表示イメージ

AutoCAD のカスタムコマンドとExcelVBA マクロを連携させたツールの作成を試みます。



■ 操作手順イメージ

① カスタムコマンド「EXPPARTSATT」を起動
② 選択オブジェクトの中からブロックの属性「設備NO、品名、寸法、品番」を読み取られ、エクセルシートにリスト表示される。
③ エクセルシート上で「設備NO、品名、寸法、品番」を任意の値に編集する。
④ カスタムコマンド「MODPARTSATT」を起動
⑤ 編集後のエクセルシートの内容に従って、該当するブロックの属性「設備NO、品名、寸法、品番」が更新･再描画される。

■ 使用するAutoCAD カスタムコマンドLISPファイル

EXPPARTSATT.lsp

MODPARTSATT.lsp

コマンド EXPPARTSATT を記述したファイル。
選択したオブジェクトからブロック参照の情報を取得し PartsList.csv に書き出す。

書き出しフォーマットは以下の通り。

通番,ブロック参照のハンドル,ブロック定義のブロック名,X座標,Y座標,設備NO属性値,品名属性値,寸法属性値,品番属性値

書き出した後、Run_GetPartsList.ps1 を起動して、Excel PartsList.xlsm を開いてマクロGetPartsList を起動する。

Run_PutPartsList.ps1 を起動して、Excel PartsList.xlsm のマクロ PutPartsList を実行させる。
（同マクロにて PartsList.csv は、編集後の属性値に書き換わる。）

EXECMOD

Excel PartsList.xlsm マクロPutPartsListの PartsList.csv の書換処理後、同マクロから起動される。PartsData.csv を参照し、
変更されたブロック参照の属性値を書き換える。

コマンド MODPARTSATT、および、EXECMOD を記述したファイル

MODPARTSATT



■ AutoLISP ファイルの他に必要なファイル

２．PartsList.xlsm

１．PartsList.csv

この2つのPowerShellスクリプトについては、AutoLISP から、直接、Excelブックのマクロを起動する方法が見つけられなかった
ため AutoLISP → PowerShell 起動 → Excelマクロ起動 という方法をとっています。VBScript を用いても同様のことが行え、
PowerShell のように処理中余計なウィンドウが表示されず、処理速度も速そうなのですが、VBSscript は2027年以降OSから削除予
定とのことで、PowerShell を使用しました。

GetPartsList

 AutoCAD側 のEXPPARTSATT コマンドにて書き出されたブロック参照の属性データ PartsList.csv を読み込み、ワークシート
にリスト表示するマクロ。後述の、 Run_GetPartsList.vbs にて起動されます。

PutPartsList

ワークシート上のブロック参照の属性データを PartsList.csv に書き出す（上書き）マクロ。後述の、 Run_PutPartsList.vbs に
て起動されます。

また、PartsList.csv の書換終了後、AutoCADカスタムコマンド EXECMOD を起動します。

AutoCAD のEXPPARTSATT コマンド、および、EXCEL PartsList.xlsm のマクロPutPartsList によってブロック参照の属性デー
タが書き込まれるデータファイル。一行のデータ構造は以下の通り。

通番,ブロック参照のハンドル,ブロック定義のブロック名,X座標,Y座標,設備NO属性値,品名属性値,寸法属性値,品番属性値

３．Run_GetPartsList.ps1 、Run_PutPartsList.ps1

属性リストを表示するためのExcel ワークブックで、下記2つのマクロを記録しておきます。

Run_GetPartsList.ps1

AutoCADの EXPPARTSATTコマンドにて起動され、ExcelブックPartsList.xlsmを開き、マクロGetPartsList を起動します。

Run_PutPartsList.ps1

AutoCADの MODPARTSATTコマンドにて起動され、ExcelブックPartsList.xlsmの マクロPutPartsList を起動します。



■ 属性値 書き出しイメージ

EXPPARTSATT コマンド
実行

① 属性データを書きだし

② Run_GetPartsList.ps1
起動

③ PartsList.xlsm を開いて
マクロGetPartsListを起動

PartsList.xlsm

PartsList.csv

Run_GetPartsList.ps1

④ 属性データを読み込み
リスト表示

〇〇.dwg



■ 属性値 書換イメージ

MODPARTSATT コマンド
実行

⑤ 属性データを読み出して
ブロック参照を書き換え

① Run_PutPartsList.ps1
起動

② マクロPutPartsListを起動

PartsList.xlsm

PartsList.csv

Run_PutPartsList.ps1

③ 属性データを上書き

〇〇.dwg
④ EXECMOD

起動

EXECMOD



■ フォルダの構造（例）

[C:/Users]
 ├── [ユーザ名]
│   ├── [AutoCAD_CustomCommands]

 │   │    ├── [Function]
 │   │    │    ├── [Get-Insert-Attributes.lsp]  Get-Insert-Attributes
 │   │    │    ├── [Change-CSV-to-List.lsp]  Change-CSV-to-List
 │   │    │    └── [Modify-Attributes.lsp]  Modify-Attributes
 │   │    ├── [AttributeEditor]
 │   │    │    ├── [EXPPARTSATT.lsp]   EXPPARTSATT
 │   │    │    ├── [MODPARTSATT.lsp]   MODPARTSATT,EXECMOD
 │   │    │    ├── [PartsList.csv]
 │   │    │    ├── [PartsList.xlsm]
 │   │    │    ├── [Run_GetPartsList.ps1]
 │   │    │    └── [Run_PutPartsList.ps1]
 │   │    └── [acaddoc.lsp]

① 個人ユーザーフォルダ直下にフォルダ [AutoCAD_CustomCommands]を置きます
② その下に、[Function] フォルダ、[AttributeEditor] フォルダ、acaddoc.lsp を置きます。
③ [Function] には、カスタム関数を置きます。
④ [AttributeEditor] には、属性編集に必要なファイル一式を置きます。

■ AutoCAD フォルダ設定

※ ｢¥...」は、下層の全てのフォルダを含むことを表します。



;■初期設定
(vl-load-com)       ; Visual LISPを有効にします。
(command-s “._COMMANDLINE”)    ; コマンドラインを表示します。
(setvar “PALETTEOPAQUE” 1)    ; パレット、コマンドウィンドウが半透明になる機能を無効化
(setvar “CMDECHO” 1)     ; コマンドエコー有効化
(setvar “DYNMODE” 3)     ; ダイナミック入力有効化
;
;■カスタム関数をロード
;
;【共通関数】       ; ファンクションをロード
 (load "Function/Change-CSV-to-List.lsp")
 (load "Function/Get-Insert-Attributes.lsp")
 (load "Function/Modify-Attributes.lsp")
;
;【属性編集コマンド】     ;コマンドをロード
 (load "AttributeEditor/EXPPARTSATT.lsp")
 (load "AttributeEditor/MODPARTSATT.lsp")

■ acaddoc.lsp の記述

acaddoc.lsp は、図面を開くタイミングで実行されるLISP ファイルです。
以下に、そのサンプルを示します。

上記は、属性編集コマンドを行うときに最低限必要なLISPファイルのみをロードしています。
初期設定部分は、不要であれば、省略しても問題ありません。



■ ExcelVBA マクロ詳細

１．GetPartsList

① シート初期化
・オートフィルタを解除
・全データ削除
・セルの書式設定（全セル：文字列）
・見出し行を作成

通番，ハンドル，ブロック名，X座標，Y座標，識別番号，品名，寸法，品番
② データ読み込み

・PartsList.csv のファイルパスを取得し(Excelブックと同じフォルダに作成とする) リードオープン
・1行目を読み込み
・区切り文字「,」にてデータを分解、配列化
・配列の要素をセルに書き込む
・データが無くなるまで繰り返し
・PartsList.csv クローズ

③ テーブル化
・セルのひとまとまりをテーブル化

④ ソート
・X座標昇順、Y座標降順 にてソート

２．PutPartsList
ワークシートの書き出しには、TextStream オブジェクトと WriteLine メソッドを使用して、1行単位でデータを書きだします。

① TextStream オブジェクトの作成
・PartsList.csv のファイルパス(Excelブックと同じフォルダに作成とする)を取得
・TextStream オブジェクト取得（上書きモード）

② データ行数の取得
・テーブルの行数を取得

③ 書き込み
・1行データを配列に代入
・同データを ｢,｣ 区切り文字列に変換
・1行書き込み
・行数分繰り返し

④ TextStream オブジェクト開放
・TextStream オブジェクトを開放する

⑤ AutoCAD コマンド EXECMOD を起動する



【Get-Insert-Attributes】関数

引数 InsName(図形名) からプロパティを取得
グループコード66を確認。( =1 であれば、属性あり）

関数 entnextにて InsNameにネストされた図形名を取得し、その図形名にリンクするプロパティを取得
図形タイプ(グループコード0) を確認。”SEQEND”であれば終わり。”ATTRIB”であれば、属性情報を取得する。

InsName のプロパティからブロック名(グループコード2) を取得
ブロック名にリンクするブロックテーブルオブジェクト名を関数 tblobjname にて取得
このテーブルオブジェクトにネストするオブジェクトのプロパティを取得
図形タイプが “ATTDEF” で、固定値、かつ、可視であれば、その固定値を取得
ネストするオブジェクトがなくなるまで繰り返す。

１．EXPPARTSLIST

・選択中のオブジェクトからブロック参照を抽出し 選択セットInsertsに代入
・Insetsが空の時は、操作者にオブジェクトの選択を促し、ブロック参照を抽出して 選択セットInsertsに代入
・PartsList.csv のファイルパスを取得し、ライトオープン
・Inserts 内 n番目のブロック参照の図形名を取得
・図形名の図形情報を取得
・図形情報から ハンドル・X座標・Y座標・ブロック名 を取得
・関数 Get-Insert-Attributes にて「ATT01 ～ ATT04｣ を取得
・1行分のデータ文字列「通番|ハンドル|ブロック名 |X座標|Y座標|ATT01| ATT02|ATT03|ATT04｣ を作成
・PartsList.csv に1行分データを書き込み
・選択セットInserts 内のブロック参照の数 これを繰り返す。
・PartsList.csv クローズ
・PartsList.xlsm のマクロ GetPartsList を起動する

■ AutoCAD カスタムコマンド(AutoLISP)詳細（１／２）

引数  ：InsName（図形名）
戻り値  ：( ( 属性名 . 属性値 ) ( 属性名 . 属性値 ) ( 属性名 . 属性値 ) ( 属性名 . 属性値 ) ････ ) 



２．MODPARTSATT

・PartsList.xlsm のマクロ GetPartsList を起動する

■ AutoCAD カスタムコマンド(AutoLISP)詳細（２／２）

【Modify-Attributes】関数

・関数 entnextにて InsNameにネストされたオブジェクト(図形名)を取得し、その図形名にリンクするプロパティを取得
・図形タイプ(グループコード0) を確認。”SEQEND”であれば終わり。”ATTRIB”であれば、属性情報を取得する。
・属性名を取得
・同名の属性名が、引数のAttList に存在するかを確認
・存在すれば、属性値を確認
・属性値が異なる場合は、同オブジェクト情報の属性値を置換して、関数 entmod を用いて書き換える
・これを “SEQEND” まで繰り返す。

引数1  ：InsName（図形名）
引数2  ：AttList ( ( 属性名 . 属性値 ) ( 属性名 . 属性値 ) ( 属性名 . 属性値 ) ( 属性名 . 属性値 ) ････ ) 

３．EXECMOD

・PartsList.csv のファイルパスを取得し、リードオープン
・1行目のデータを読む
・データから属性名と属性値のリストを作成する

( ( 属性名 . 属性値 ) ( 属性名 . 属性値 ) ( 属性名 . 属性値 ) ( 属性名 . 属性値 ) )
・関数 ModAttributes を用いて 該当するハンドルを持つ図形の属性を更新する
・これをデータの行数分繰り返す。
・PartsList.csv をクローズ



１．Run_GetPartsList

・PartsList.xlsm が開かれていなければ開く
・マクロGetPartsList を起動する

■ PowerShell 詳細

２．Run_PutPartsList

・PartsList.xlsm が開かれていなければ開く→ 不要？
・マクロPutPartsList を起動する

■ サンプル動画

クリック

動画が再生されます

属性を編集する1

クリック

動画が再生されます

属性を編集する2

・それぞれの備品をブロックとして定義、｢設備NO｣、｢品名｣、｢寸法｣、｢品番｣の
属性を持たせる。

・同種の設備毎に設備NOを付与、併せて、品名、寸法、品番を記録していく

・すべての備品に、属性を登録完了
・全ブロック定義を選択して、リスト表示
・オートフィルタを用いて、表示変更することで、品名とその員数などを確認できる。

属性をエクセルにてリスト表示することで、以下に示すメリットが生じると考えます。

・属性値の登録･編集作業が、エクセルの持つ多彩な機能をて行えるため、作業効率が向上します。
・エクセルのオートフィルタなどの機能を用いて、設備の種類・員数などの確認が容易にできるようになります。
・さらに、ExcelVBAを用いて「設備の項目と員数などをまとめた構成表を作成する」、「設備の仕入れ価格･仕入先などを記憶したデータ

ベースとリンクさせて手配書･見積書を作成する」など、アイデア次第で有用なツールの作成が可能です。

しかしながら、以下のデメリットも考えられます。

・使用するブロック定義を一元的に管理･徹底する運用を前提としないと、かえって誤り･混乱を生じさせる恐れがある。

■ 本ツール作成の効果(メリット／デメリット)

※ 本動画は、Excelマクロの起動に、VBScript を用いたときのもので、PowerShell の場合とは少し処理中画面が異なりますが機能は同じです。

https://kibougaoka-yokohama.site/wp-content/uploads/autocad/sub-contents/video/属性エクセル編集サンプル1.mp4
https://kibougaoka-yokohama.site/wp-content/uploads/autocad/sub-contents/video/属性エクセル編集サンプル2.mp4


■ ExcelVBA マクロソース（１／５）

１．GetPartsList

Declare PtrSafe Function FindWindow Lib "user32" Alias "FindWindowA" ( _
    ByVal lpClassName As String, ByVal lpWindowName As String) As LongPtr
Declare PtrSafe Function SetWindowPos Lib "user32" ( _
    ByVal hWnd As LongPtr, _
    ByVal hWndInsertAfter As LongPtr, _
    ByVal X As Long, _
    ByVal Y As Long, _
    ByVal cx As Long, _
    ByVal cy As Long, _
    ByVal wFlags As Long) As Long
Public Const HWND_TOPMOST = -1
Public Const SWP_NOMOVE = &H2
Public Const SWP_NOSIZE = &H1
Public Const SWP_SHOWWINDOW = &H40

Public Sub GetPartsList()

'■ シートの初期化
'ウィンドウを最前面に
Dim hWnd As LongPtr

    hWnd = FindWindow("XLMAIN", Application.Caption)
    SetWindowPos hWnd, HWND_TOPMOST, 0, 0, 0, 0, SWP_NOMOVE Or SWP_NOSIZE Or SWP_SHOWWINDOW
    'ブックをアクティブに

ThisWorkbook.Activate
    '最小化されていたら元に戻す

If ActiveWindow.WindowState = xlMinimized Then
        Application.WindowState = xlNormal
    End If

  （つづく）



■ ExcelVBA マクロソース（２／５）

   （つづき）

'■ Sheet(1)を表示
Worksheets(1).Activate

    'シートの表示を初期化
Application.Goto Reference:=Range("A1"), Scroll:=True     'セル(1 , 1)が表示されるようスクロール
With ActiveSheet

        If .FilterMode = True Then .ShowAllData               'オートフィルタ解除
.Cells.Delete                                         '全セル クリア
.Cells.NumberFormat = "@"                             '全セル 書式→文字列

End With
'
'■ PartsList.csv オープン

Dim Filename As Variant
    Dim CsvNo As Integer
    Filename = ActiveWorkbook.Path & "¥PartsList.csv"
    CsvNo = FreeFile
    Open Filename For Input As #CsvNo
'
'■ ブロック参照属性データロード

Dim LineDataStr As String
    Dim LineDataAry() As String
    Dim i As Long
    i = 1
    Do Until EOF(CsvNo)
        Line Input #CsvNo, LineDataStr
        LineDataAry = Split(LineDataStr, ",")
        Range(Cells(i, 1), Cells(i, UBound(LineDataAry) + 1)) = LineDataAry
        i = i + 1
    Loop
'
'■ PartsList.csv クローズ

Close #CsvNo
‘

    （つづく）

１．GetPartsList



■ ExcelVBA マクロソース（３／５）

   （つづき）

'■ テーブル化
Dim TBL As ListObject

    Set TBL = ActiveSheet.ListObjects.Add(xlSrcRange, Cells(1, 1).CurrentRegion, , xlYes)
    With TBL
        .TableStyle = "TableStyleMedium15"
        .ShowTableStyleRowStripes = False
    End With
'
'■ 列幅調整

TBL.Range.Columns.AutoFit
'
'■ ソート

If TBL.DataBodyRange Is Nothing Then Exit Sub
    
    Dim Clm1 As Long, Clm2
    Clm1 = TBL.ListColumns("X座標").Range.Column
    Clm2 = TBL.ListColumns("Y座標").Range.Column
    With ActiveSheet.Sort.SortFields
        .Clear
        .Add Key:=Cells(1, Clm1), SortOn:=0, Order:=1, DataOption:=1
        .Add Key:=Cells(1, Clm2), SortOn:=0, Order:=2, DataOption:=1
    End With
    With ActiveSheet.Sort
        .SetRange TBL.Range
        .Header = xlYes
        .MatchCase = True
        .Orientation = xlTopToBottom
        .SortMethod = xlPinYin
        .Apply
    End With

End Sub

   （つづき）

１．GetPartsList



■ ExcelVBA マクロソース（４／５）

２．PutPartsList

Public Sub PutPartsList()
    
    Dim CsvFileName As Variant
    Dim FileSysObj As New FileSystemObject
    Dim RowsTextStream As TextStream
    Dim minRow As Long, maxRow As Long
    Dim minClm As Long, maxClm As Long
    Dim i As Long
    
    CsvFileName = ActiveWorkbook.Path & "¥PartsList.csv"
    
    minRow = 1
    minClm = 1
    With Cells(1, 1).CurrentRegion
        maxRow = .Rows.Count
        maxClm = .Columns.Count
    End With

    Set RowsTextStream = FileSysObj.CreateTextFile(Filename:=CsvFileName, Overwrite:=True)
    For i = minRow To maxRow
        RowsTextStream.WriteLine LineData(i, minClm, maxClm)
    Next

    RowsTextStream.Close

    Set RowsTextStream = Nothing
    Set FileSysObj = Nothing

    （つづく）



   （つづき）
    Dim acadApp As Object
    Dim acadDoc As Object
    Set acadApp = GetObject(, "AutoCAD.Application.25")
    Set acadDoc = acadApp.ActiveDocument
   
    Run_CMD = "EXECMOD" & vbCrLf
    acadDoc.SendCommand Run_CMD
   
    Set acadDoc = Nothing
    Set acadApp = Nothing

End Sub

Public Function LineData (ByVal xRow As Long, minClm As Long, maxClm As Long) As String

    Dim buf As Variant
    Dim xClm As Long
    
    LineData = ""
    For xClm = minClm To maxClm
        buf = Cells(xRow, xClm)

If IsNumeric(buf) Then buf = CStr(buf)
If LineData = "" Then

            LineData = buf
        Else
            LineData = LineData & "," & buf         '
        End If
    Next

End Function

■ ExcelVBA マクロソース（５／５）

２．PutPartsList



１．EXPPARTSATT.lsp

■ AutoLISP ソース（１／１０）

(Defun C:ExpPartsAtt ( / )
;
(vl-load-com)
(princ "¥nStart[ExpPartsAtt]¥n")
(setq startmsec (getvar "MILLISECS"))
(setvar "CMDECHO" 0)
(setvar "DYNMODE" 3)
;
(setq CommandPath (findfile "AutoCAD_CustomCommands"))
(setq CsvFilePath (strcat CommandPath "¥¥AttributeEditor¥¥PartsList.csv"))
(setq PSFilePath (strcat CommandPath "¥¥AttributeEditor¥¥Run_GetPartsList.ps1"))
;
;オブジェクト選択
(setq Inserts (ssget "_I" '((0 . "INSERT"))))
(if
 (= Inserts nil)
 (setq Inserts (ssget '((0 . "INSERT"))))
);if
;
;PartsList.csv オープン
(setq CSV (open CsvFilePath "w"))
;見出し行書き出し
(write-line "番号,Handle,ブロック名,X座標,Y座標,設備NO,品名,寸法,品番" CSV)
;
;   （つづく）



■ AutoLISP ソース（２／１０）

;    （つづき）
;
;属性取得＆書き出し
(if
 Inserts
 (progn

(setq InsLen (sslength Inserts))
  (setq InsIdx 0)
  (repeat InsLen
   (setq No (itoa (1+ InsIdx)))
   ;ブロック参照のプロパティを取得

(setq EntName (ssname Inserts InsIdx))
   (setq Properties (entget EntName))
   (setq Handle (cdr (assoc 5 Properties)))
   (setq Coord (cdr (assoc 10 Properties)))
   (setq X (car Coord))
   (setq Y (cadr Coord))
   (setq BlkName (vla-get-Effectivename (vlax-ename->vla-object EntName)))
   (if
    ;関数Get-Insert-Attributesにて 属性を取得

(setq Attributes (Get-Insert-Attributes EntName))
    (progn
     ;1行分のデータリストを作成

(setq DataList (list Handle BlkName X Y
      (cdr(assoc "設備NO" Attributes))
      (cdr(assoc "品名" Attributes))
      (cdr(assoc "寸法" Attributes))
      (cdr(assoc "品番" Attributes))
     ))
     ;リスト内の nil を "－－－" に差し替え

(setq DataList (subst "－－－" nil DataList))
     (setq DataList (subst "－－－" "" DataList))
;
;    （つづく）

１．EXPPARTSATT.lsp



    ;1行分のカンマ区切り文字列を作成(同時にリスト内の数値は文字列に変換)
     (setq WriteData No)
     (setq LstLen (length DataList))
     (setq LstIdx 0)
     (repeat LstLen
      (setq Data (nth LstIdx DataList))
      (if
       (numberp Data)
       (setq Data (rtos Data 2 2))
      );if
      (setq WriteData (strcat WriteData "," Data))
      (setq LstIdx (1+ LstIdx))
     );repeat
     ;1行分のデータを書き込み

(write-line WriteData CSV)
    );progn
   );if
   (setq InsIdx (1+ InsIdx))

);repeat
 );progn
);if
;エクセルマクロ[GetPartsList]起動
(startapp "powershell" (strcat "-ExecutionPolicy Bypass -File ¥"" PSFilePath "¥""))
;PartsList.csv クローズ
(close CSV)
;
(setvar "CMDECHO" 1)
(setq endmsec (getvar "MILLISECS"))
(princ "¥n*** end")(princ (- endmsec startmsec))
(princ)
)

■ AutoLISP ソース（３／１０）

   （つづき）

１．EXPPARTSATT.lsp



２．MODPARTSATT.lsp

■ AutoLISP ソース（４／１０）

(Defun C:ModPartsAtt ( / CommandPath VbsFilePath)
;
(princ "¥n[Run_PutPartsList]¥n")
(setq startmsec (getvar "MILLISECS"))
(setvar "CMDECHO" 0)
(setvar "DYNMODE" 3)
(command-s "._COMMANDLINE")
;
(setq CommandPath (findfile "AutoCAD_CustomCommands"))
(setq PSFilePath (strcat CommandPath "¥¥AttributeEditor¥¥Run_PutPartsList.ps1"))
;
;エクセルマクロ[PutPartsList]起動
(startapp "powershell" (strcat "-ExecutionPolicy Bypass -File ¥"" PSFilePath "¥""))
;
(setvar "CMDECHO" 1)
(princ)
);defun
;----------------------------------------------------------------------------------------------------------
(Defun C:EXECMOD ( / CommandPath CsvFilePath CSV Hder HdLst Dat DatLst HdlIdx Hdl AttLst AttLen Idx AttName AttDat 
AttDotPear InsName)
;
(princ "¥nStart[Mod]¥n")
(setq startmsec (getvar "MILLISECS"))
(setvar "CMDECHO" 0)
(setvar "DYNMODE" 3)
(command-s "._COMMANDLINE")
;
(setq CommandPath (findfile "AutoCAD_CustomCommands"))
(setq CsvFilePath (strcat CommandPath "¥¥AttributeEditor¥¥PartsList.csv"))
;



;PartsList.csv オープン
(setq CSV (open CsvFilePath "r"))
;
(setq Hder (read-line CSV))
(setq HdLst (Change-Csv-to-List Hder))
(while
 (and
  (setq Dat (read-line CSV))
  (setq DatLst (Change-Csv-to-List Dat))
 );and
 ;
 (setq HdlIdx (vl-position "Handle" HdLst))
 (setq Hdl (nth HdlIdx DatLst))
 (setq AttLst nil)
 (setq AttLen 4)
 (setq Idx 5)
 (repeat AttLen
  (setq AttName (nth Idx HdLst))
  (setq AttDat (nth Idx DatLst))
  (if
   (/= AttDat "－－－")
   (progn
    (setq AttDotPear (cons AttName AttDat))
    (setq AttLst (append AttLst (list AttDotPear)))
   );progn
  );if
  (setq Idx (1+ Idx))
 );repeat
 (setq InsName (handent Hdl))
 (Modify-Attributes InsName AttLst)
);while

■ AutoLISP ソース（５／１０）

２．MODPARTSATT.lsp

   （つづき）

   （つづく）



;
(close CSV)
;
(setvar "CMDECHO" 1)
(setq endmsec (getvar "MILLISECS"))
(princ "¥n*** end ")(princ (- endmsec startmsec))(princ " msec")
(princ)
);defun
;----------------------------------------------------------------------------------------------------------

■ AutoLISP ソース（６／１０）

２．MODPARTSATT.lsp

   （つづき）



■ AutoLISP ソース（７／１０）

３．Get-Insert-Attributes.lsp

;INSERTの属性(ATTRIB)のドットペアリストを返す関数
;
(Defun Get-Insert-Attributes (InsName / InsProperties AttFlg NextName NextProperties Type AttName AttValue AttList)
 (setq AttList nil)
 (setq InsProperties (entget InsName))
 (if

(= (cdr (assoc 66 InsProperties)) 1) ;属性あり
  (progn
   ;■[INSERT]内を検索

(setq NextName InsName)
   (while
    (and
     (setq NextName (entnext NextName))
     (setq NextProperties (entget NextName))
     (setq Type (cdr (assoc 0 NextProperties)))
     (/= Type "SEQEND")
    );and
    (if

(and
      (= Type "ATTRIB")
      (or
       (not (assoc 60 NextProperties))
       (= (cdr (assoc 60 NextProperties)) 0)
      )or
     );and
     (progn
      (setq AttName (cdr (assoc 2 NextProperties)))
      (setq AttValue (cdr (assoc 1 NextProperties)))
      (setq AttList (append AttList (list (cons AttName AttValue))))
     ):progn
    );if
   );while

   （つづく）



  ;■[BLOCK]内を検索
(setq InsProperties (entget InsName))

   (setq BlkName (cdr (assoc 2 InsProperties)))
   (setq NextBlk (tblobjname "Block" BlkName))
   (while 

(and
     (setq NextBlk (entnext NextBlk))  ;ネストされたオブジェクトあり
     (setq BlkProperties (entget NextBlk))
     (setq Type (cdr (assoc 0 BlkProperties)))
    );and
    (if

(and
                   (= Type "ATTDEF")
                   (= (boole 1 (cdr (assoc 70 BlkProperties)) 2) 2)
                   (or
       (not (assoc 60 BlkProperties))
       (= (cdr(assoc 60 BlkProperties)) 0)
      )
     );and
     (progn
      (setq AttName (cdr(assoc 2 BlkProperties)))
      (setq AttValue (cdr(assoc 1 BlkProperties)))
      (setq AttList (append AttList (list (cons AttName AttValue))))
             );progn
    );if
   );while
  );progn
 );if
;
AttList
)

■ AutoLISP ソース（８／１０）

３．Get-Insert-Attributes.lsp

   （つづき）

   （つづく）

■ AutoLISP ソース（９／１０）

;引数[Str] ：カンマ区切り文字列
;戻り値[Lst] ：引数をカンマにて分割したリスト
;
(Defun Change-Csv-to-List (Str / Lst Pos Dat)

(setq Lst nil)
(while

(setq Pos (vl-string-search "," Str)) ;Pos = カンマの位置(先頭は0)
;
(setq Dat (substr Str 1 Pos))  ;Dat = 先頭からPosまでの文字列
(setq Str (substr Str (+ Pos 2))) ;Str = StrからDatと"," を除いた文字列)
(setq Lst (append Lst (list Dat))) ;LitにDatを追加していく

);while
(setq Lst (append Lst (list Str)))  ;Strの残り部分(","を含まない末尾)をリストに追加
Lst

);defun

３．Get-Insert-Attributes.lsp

   （つづき）



■ AutoLISP ソース（１０／１０）

;引数1：InsName  INSERTの図形名
;引数2：AttValueList ドットペアリスト ((属性名 . 属性値) (属性名 . 属性値) ･･･ )
;戻り値：なし
;
(Defun Modify-Attributes (InsName AttValueList / BlkOrgName NextName NextProperties NextType AttName NewAttValue)
;
 (setq BlkOrgName (vla-get-Effectivename (vlax-ename->vla-object InsName)))
 (setq NextName InsName)
 (while
  ;ネストされた図形名が存在し、そのタイプが"SEQEND"ではない

(and
   (setq NextName (entnext NextName))
   (setq NextProperties (entget NextName))
   (setq NextType (cdr (assoc 0 NextProperties)))
   (/= NextType "SEQEND")
  );and
  (if
   ;属性情報の属性名と同じ属性名が引数2:AttValueList に存在する

(and
    (= NextType "ATTRIB")
    (setq AttName (cdr (assoc 2 NextProperties)))
    (setq OldValue (cdr (assoc 1 NextProperties)))
    (assoc AttName AttValueList)
    (setq NewValue (cdr (assoc AttName AttValueList)))
    (/= NewValue OldValue)
    (setq NextProperties (subst (cons 1 NewValue) (cons 1 OldValue) NextProperties))
   );and
   (entmod NextProperties)
  );if
 );while
 (princ)
)

４．Modify-Attributes.lsp



１．Run_GetPartsList

■ PowerShellソース（１／２）

$scriptPath = $MyInvocation.MyCommand.Path
$folderPath = Split-Path -Parent $scriptPath
$excelFileName = "PartsList.xlsm"
$macroName = "GetPartsList"
$fullExcelPath = Join-Path $folderPath $excelFileName
# Excel アプリケーション取得または起動
$excelApp = $null
try {
    $excelApp = [Runtime.InteropServices.Marshal]::GetActiveObject("Excel.Application")
} catch {
    $excelApp = New-Object -ComObject Excel.Application
    $excelApp.Visible = $true
}
# ブックを開くまたは既存のブックを参照
$workbook = $null
foreach ($wb in $excelApp.Workbooks) {
    if ($wb.FullName -eq $fullExcelPath) {
        $workbook = $wb
        break
    }
}
if (-not $workbook) {
    $workbook = $excelApp.Workbooks.Open($fullExcelPath)
}
# マクロ実行
$excelApp.Run("$excelFileName!$macroName")
# COM オブジェクトの解放
[System.Runtime.InteropServices.Marshal]::ReleaseComObject($workbook) | Out-Null
[System.Runtime.InteropServices.Marshal]::ReleaseComObject($excelApp) | Out-Null
[GC]::Collect()
[GC]::WaitForPendingFinalizers()



$MyFolderPath = Split-Path -Parent $MyInvocation.MyCommand.Definition
$ExcelBookName = "PartsList.xlsm"
$Macro = "PutPartsList"
$ExcelBookFullName = Join-Path $MyFolderPath $ExcelBookName

# Excel アプリケーションを取得または作成
try {
    $ExcelAppObj = [Runtime.InteropServices.Marshal]::GetActiveObject("Excel.Application")
    $ExcelOpened = $false
    try {
        $ExcelBookObj = $ExcelAppObj.Workbooks.Item($ExcelBookName)
    } catch {
        $ExcelBookObj = $ExcelAppObj.Workbooks.Open($ExcelBookFullName)
        $ExcelOpened = $true
    }
} catch {
    $ExcelAppObj = New-Object -ComObject Excel.Application
    $ExcelAppObj.Visible = $true
    $ExcelBookObj = $ExcelAppObj.Workbooks.Open($ExcelBookFullName)
    $ExcelOpened = $true
}

# マクロの実行
$macroFullName = "$ExcelBookName!$Macro"
$ExcelAppObj.Run($macroFullName)

# 解放処理（COM オブジェクトのクリーンアップ）
[System.Runtime.InteropServices.Marshal]::ReleaseComObject($ExcelBookObj) | Out-Null
[System.Runtime.InteropServices.Marshal]::ReleaseComObject($ExcelAppObj) | Out-Null
[GC]::Collect()
[GC]::WaitForPendingFinalizers()

２．Run_PutPartsList

■ PowerShellソース（２／２）

